 Apiola/Pajukangas

Mat-1.414 Matematiikan peruskurssi V2 kevät 2002
`http://www.math.hut.fi/teaching/v/3/H/`

Laskuharjoitus 12 (viikko 17 , 24–26.4.2002)

Tämä vihoviimeinen harjoitus sisältää myös parin edellisen harjoituskerran aihepiirejä, erityisesti avaruusintegraaleja ja muuttujanvaihdoksia.

Muistakaapa "kansalaisvelvollisuus", **palaute**, kaavake ilmestyy kurssisivulle tuotapikaa. Palautteella voit vaikuttaa myös syksyn V3- kurssin toteutustapaan.

Malliratkaisuja tulee, mahdollisesti yhteenvetoja ja ainakin välikoealueen tärkennusta (aika kattava kuvaus on jo).

Pysy siis linjoilla!

Alkuviikko (AV)

1. Olkoon

$$F(t) = \int_t^{t^2} \frac{\sin \frac{x}{t}}{x} dx, \quad t \neq 0.$$

Laske $F'(t)$
Vast: $\frac{\sin t}{t}$

2. Osoita, että integraali

$$\int_A \frac{da}{x+y}$$

suppenee, kun A on joukko $\{(x,y) | \ 0 \leq y \leq \frac{1}{x}, \ \ x \geq 1\}$

3. Osoita, että integraali

$$\int_A \frac{da}{xy}$$

hajaantuu, kun A on suoran $y = x$ ja paraabelin $y = x^2$, $0 \leq x \leq 1$ rajoittama alue.

4. Gaussin kvadratuurissa (numeerisessa integroinnissa) etsitään muotoa $\sum_{i=1}^n w_i f(x_i)$ olevaa approksimaatiota integraalille $\int_{-1}^1 f(x)dx$ siten, että mahdollisimman korkea-asteiset (2n-1) polynomit integroituisivat tarkasti.

Muodosta kahden pisteen ($n = 2$) Gaussin kaava, eli ratkaise solmut x_1, x_2 ja painot w_1, w_2.

Vihje: Kirjoita vaatimus tarkasta integraalista monomeille $1, x, x^2, x^3$ (tämän kirjoitin taululle luennolla) ja ratkaise yhtälösysteemistä solmut ja painot. Lisävihje: Ratkaisu muuttuu oikein helpoksi, jos käytät hyväksesi luonnollisia symmetrioita.

Huom! Tämä tehtävä **ei edellytä sitten minkäänlaisia esitietoja** yhtään mistään!

5. Muodosta edellisen tehtävän kaavan avulla Gaussin tulokaava yksikköneliölle $[-1,1] \times [-1,1]$. (Jos jostain syystä et laskenut edellistä, niin etsi kaava viitteistä.)

Laske sen avulla approksimaatio integraalille:

$$\int_{1.4}^{2} \int_{1}^{1.5} \ln(x + 2\,y)\, dy\, dx$$

Kuinka monta oikeaa numeroa saat. Tässä Maplelasku 9:llä numerolla.

```
> int(int(ln(x+2*y),y=1..3/2),x=14/10..2);
                                    321          121            289
                41/5 ln(5) - 9/20 - --- ln(2) - --- ln(11) + --- ln(17
                                    25           25             100

> evalf(%);
                                           .429554525
```

Ohje: Suorita muunnos yksikköneliöltä ao. suorakulmiolle. Muunnoksen saat vaikkapa niin, että muunnat ensin lineaarikuvauksella (diagonaalimatriisilla) yksikköneliön oikean muotoiseksi suorakulmioksi ja sitten teet siirron.

6. Usein käytännön sovelluksissa esiintyvä tehtävä on integrointi kolmion yli. (FEM-laskenta).

Laskettavana olkoon $\int_\Delta f(x,y)da$, missä $\Delta = \{(x,y) | \ \ 0 \leq x \leq 1, \ \ 0 \leq y \leq x\}$.

(a) Suorita muuttujanvaihto $u = x, v = \frac{y}{x}$ ja totea, että se muuntaa neliön $[0,1]\times[0,1]$ kolmiolle Δ melkein kaikkialla bijektiivisesti (O poissuljettuna)

(b) Osoita, että integraali saadaan lasketuksi kaavalla:

$$\int_\Delta f(x,y)da = \int_0^1 du \int_0^1 f(u,uv)udv$$

(c) Tee vielä muunnos, joka muuntaa tämän integraaliksi yli yksikköneliön $[-1,1]\times[-1,1]$.

Ohjeita

Integraalin derivointi

Olkoon $f : \mathbb{R}^2 \to \mathbb{R}$ jatkuva ja olkoon osittaisderivaatta $f_t(t,x)$ olemassa ja jatkuva, kun $x \in [a,b]$ ja $t \in B$, missä B on jokin $\mathbb{R}$:n väli. Olkoon

$$F(t) = \int_{a(t)}^{b(t)} f(t,x)dx,$$

missä $a(t)$ ja $b(t)$ ovat derovoituvia. Tällöin

$$F'(t) = f(t,b(t))b'(t) - f(t,a(t))a'(t) + \int_{a(t)}^{b(t)} f_t(t,x)dx.$$

Loppuviikko (LV)

1. Laske AV teht. 5 korkeamman asteisilla Gaussin kaavoilla (kuin AV). Tutustu samalla Gauss-integrointifunktioihin, joita on tiedostossa `../maple/v202.mpl`. Opiskele `L/numint`j `.html`, $j = 1,2,3$.

   Helpointa lienee soveltaa funktiota `Gaussxy`. Myös funktio `Gaussnelio` soveltuu mainiosti. Muista generoida ensin taulukot tyyliin:

   ```
   gtaulux:=gausstaulukko(5); gtauluy:=gausstaulukko(4);
   ```

   Selvittele funktion laskentamäärien (= tulotaulukon koon) suhdetta tuloksen tarkkuuteen.

2. Laske $\int_\Delta f(x,y)da$, missä $\Delta = \{(x,y)| \quad 0 \le x \le 1, \quad 0 \le y \le x\}$. (vrt. AV teht. 6). ja $f(x,y) = e^{-x^2y^2}$.

   Laske tehtävä käyttäen hyväksi AV 6:n muunnoksia ja funktiota `Gaussnelio`. Piirrä myös yksikköneliön $[-1,1]\times[-1,1]$ Gausspisteiden kuvapisteet kolmiossa.

   (Ohjeita: harj12ohje.mws)

3. Opiskele tiedoston `L/numint3.html` (tai `mws`) kartioesimerkki ja laske loppuun kartion massakeskipiste. Käytä lieriökoordinaatteja.

   Kyseessä on kartion $z = \sqrt{x^2+y^2}$ ja tason $z = 2$ rajoittama kappale, jonka tiheys on kaikenlisäksi $\rho(x,y,z) = \sqrt{x^2+y^2}$

4. Jatketaan vielä toruksien parissa. Pinnan parametriesityshän on

   $$\begin{cases} x = (a + b\cos v)\cos u \\ y = (a + b\cos v)\sin u \\ z = b\sin v \end{cases}$$

   Jos lasketaan avaruusintegraali yli toruksen, on kätevää esittää kappale "toruskoordinaateissa", ts. otetaan b:n paikalle kolmas muuttuja w.

   Laske toruksen tilavuus ja toruksen puolikkaan : $0 \le u \le \pi$ keskiö. (v ja w käyvät luonnollisesti läpi luonnolliset arvoalueensa)

5. Tiedostossa `.../maple/v202.mpl` on toteutettuna luennolla hahmoteltu *Monte Carlo*-menetelmä, listaus myös alla.

   Tuntuman saamiseksi kannattaa verrata laskentetyötä aivan tavalliseen Riemannin summaan. Onko satunnaisuudessa taikaa? Tästä syystä kirjoitin myös Riemann2-funktion, jossa ympäröivä suorakulmio jaetaan tasavälisesti.

   Tutki 1-ympyrän pinta-alan laskentaa ympäröimällä ympyrä yksikköneliöllä $[-1,1]\times[-1,1]$.

   Laske likiarvo π:lle molemmilla tavoilla. Huomaa, että Riemannin N vastaa Monte Carlon arvoa N^2. Kokeilu voisi sujua tähän tapaan:

   ```
   N:=10:[MonteCarlo2d(1,(x,y)->x^2+y^2,T,N^2),
         Riemann2d(1,(x,y)->x^2+y^2,T,N)];
   N:=20: ...
   ```

6. Laske tehtävän 4 toruksen puolikkaan tilavuudelle ja keskiölle likiarvoja MonteCarlo3d:llä ja vertaa tarkkoihin. Voit ottaa vaikka $a = 2, \quad b = 1$.

   Ehkä kannattaa tehdä kohtuullisen suurella N:llä useampia laskuja ja ottaa niiden keskiarvo.

Ohjeita

Numeerinen integrointi

Opiskele `L/numint1.html, numint2.html, numint3.html`. Siellä näkyy tarinan juoni, joka johti mm. alla kerättyihin koodeihin. Ne on poimittu tiedostosta `maple/v202.mpl`, joka siis on syytä lukea Maple-istuntoon: `read("v202.mpl");`

```
# Yksiulotteinen:
Gaussab:=(taulu,f,a,b)->
Gaussint(taulu,t->f(1/2*t*b-1/2*t*a+1/2*b+1/2*a)*(1/2*b-1/2*a));

# Tasointegraali yli neliön [-1,1] x [-1,1]
#
Gaussnelio:=proc(gtaulux,gtauluy,f)
# Suorita ensin gtaulux:=gausstaulukko(m);
# gtauluy:=gausstaulukko(n);
 local sx,wx,sy,wy,i,j,m,n;
 sx:=gtaulux[1]; wx:=gtaulux[2];
 sy:=gtauluy[1]; wy:=gtauluy[2];
 m:=nops(sx); n:=nops(sy);
 sum(wx[i]*sum(wy[j]*f(sx[i],sy[j]),j=1..n),i=1..m);
end:

# Kaksiulotteinen iteroitu integraali
#
Gaussxy:=(gtaulux,gtauluy,f,a1,b1,a2,b2)->
Gaussab(gtaulux,x->Gaussab(gtauluy,y->f(x,y),a2(x),b2(x)),a1,b1);

# Suorita ensin gtaulux:=gausstaulukko(m);gtauluy:=gausstaulukko(n);
#   x-projisoituva alue
# a1 ja b1 ovat x-vakiorajat
# a2 ja b2 ovat yhden muutt. funktioita
# Alue: a1 <= x <= b1, a2(x) <= y <= b2(x)

# Kolmiulotteinen iteroitu int.:
#
Gaussxyz:=(gtaulux,gtauluy,gtauluz,f,a1,b1,a2,b2,a3,b3)->
Gaussxy(gtaulux,gtauluy,(x,y)->Gaussab(gtauluz,z->
f(x,y,z),a3(x,y),b3(x,y)),a1,b1,a2,b2);

# Suorita ensin gtaulux:=gausstaulukko(m);gtauluy:=gausstaulukko(n);
#               gtauluz:=gausstaulukko(p);
#   xy-projisoituva alue
# a1 ja b1 ovat x-vakiorajat
# a2 ja b2 ovat yhden muutt. funktioita
# a3 ja b3 ovat 2 muutt. fkt.
# Alue: a1 <= x <= b1, a2(x) <= y <= b2(x), a3(x,y) <= z <= b3(x,y)


# Monte Carlo-menetelmä
#
#  Tehdään ensin 2d-versio, jossa integroimisalue ympäröidään
#  suorakulmiolla T.
#  Ehto alueeseen kuulumiselle esitetään tyyliin
#    g(x,y) <= 1
#  Tätä pitää tarpeen mukaan modifioida.
#
MonteCarlo2d:=proc (f, g, T, N)
local k, summa, lkm, xv, yv, x, y, lx, ly, A;
xv := T[1]; yv := T[2];
lx := xv[2]-xv[1]; ly := yv[2]-yv[1];
A := lx*ly; summa := 0.; lkm := 0;
for k to N do
  x := xv[1]+lx*satu(); y := yv[1]+ly*satu();
  if evalf(g(x,y)) <= 1. then
    summa := summa+f(x,y); lkm := lkm+1
  end if
end do;
A*summa/N
end proc:
```

```
# Argumentit:  f -- integroitava funktio, esim: f:=(x,y)->x*y;
#              g -- ehto, esim: g:=(x,y)->x^2+y^2;
#              T -- ympäröivä suuntaissärmiö, esim:
#                        T:=[[-1,1],[-1,1]];
#              N -- Kuinka monta pistettä arvotaan.
# Alustus:
#              satu:=rand(0..10^6)/10.^6;
```

Vertailun vuoksi Riemannin summa

```
# Samat argumentit muuten, mutta Riemannin N vastaa Monte Carlon N^2
# Tässä ei mitään satunnaislukualustusta.

Riemann2d:=proc(f,g,T,N)
local k,summa,lkm,xv,yv,x,y,lx,ly,A,X,Y,i,j;
xv:=T[1]; yv:=T[2];
lx:=xv[2]-xv[1]; ly:=yv[2]-yv[1];
A:=lx*ly;
X:=linspace(op(xv),N); Y:=linspace(op(yv),N);
# xy:=seq(seq([X[i],Y[j]],j=1..N),i=1..N);
summa:=0.; lkm:=0;
  for i to N-1 do
   for j to N-1 do
    if evalf(g(X[i],Y[j])) <= 1. then
      summa:=summa+f(x,y); lkm:=lkm+1;
    end if;
   end do;
  end do;
A*summa/N^2;
end:

# Sitten 3d-Monte Carlo (vrt. luennolla)

MonteCarlo3d:=proc(f,g,T,N)
local k,summa,lkm,xv,yv,zv,x,y,z,lx,ly,lz,V;
xv:=T[1]; yv:=T[2]; zv:=T[3];
lx:=xv[2]-xv[1]; ly:=yv[2]-yv[1];lz:=zv[2]-zv[1];
V:=lx*ly*lz;
summa:=0.; lkm:=0;
for k to N do
  x:=xv[1]+lx*satu();
  y:=yv[1]+ly*satu();
  z:=zv[1]+lz*satu();
  if evalf(g(x,y,z)) <= 1.
    then summa:=summa+f(x,y,z); lkm:=lkm+1;
  end if;
end do;
V*summa/N;
end:

# Argumentit:  f -- integroitava funktio, esim: f:=(x,y,z)->x*y*z;
#              g -- ehto, esim: g:=(x,y,z)->z^2+(sqrt(x^2+y^2)-3)^2;
#              T -- ympäröivä suuntaissärmiö, esim:
#                        T:=[[1,4],[-3,4],[-1,1]];
#              N -- Kuinka monta pistettä arvotaan.
# Alustus:
#              satu:=rand(0..10^6)/10.^6;
```